ΣΗΜΕΙΩΣΕΙΣ ΥΠΟΛΟΓΙΣΤΙΚΗΣ ΦΥΣΙΚΗΣ

Κων/νος Κόκκοτας
Τμήμα Φυσικής, ΑΠΘ

18 Μαρτίου 2011

Περιεχόμενα

0.1 ΔΙΑΦΟΡΙΚΕΣ ΕΞΙΣΩΣΕΙΣ & Mathematica

Η Mathematica έχει τη δυνατότητα αριθμητικής αλλά και αναλυτικής επίλυσης διαφορικών εξισώσεων. Αρχικά θα μελετήσουμε την αναλυτική επίλυση ΔΕ αλλά στη συνέχεια θα επικεντρώσουμε την προσοχή μας στην αριθμητική επίλυση ΔΕ.

0.1.1 Αναλυτική Επίλυση ΔΕ

Ας θεωρήσουμε τη διαφορκή εξίσωση του αρμονικού ταλαντωτή

In[1]:= **eq1 = y''[t] + w0^2 y[t] == 0**

με αρχικές συνθήκες

In[2]:= **initial = {y[0] == A, y'[0] == B}**

προσθέτοντας τις αρχικές συνθήκες στην αρχική ΔΕ δημιουργούμε τη λίστα:

In[3]:= **eq2 = Append[initial, eq1]**

η αναλυτική λύση δίνεται με εφαρμογή της εντολής **DSolve**

In[4]:= **eq3 = DSolve[eq2, y[t], t]// Flatten // ExpandAll**

Out[4]= $\left\{ \texttt{y[t]} \to \texttt{A cos[t w0]} + \frac{\texttt{B sin[t w0]}}{\texttt{w0}} \right\}$

η εντολή **Flatten** εξαλείφει ένα ζεύγος αγκίστρων ({}).

In[5]:= **eq3 /. {Rule → Equal}**

Out[5]= $\left\{ \texttt{y[t] == A cos[t w0]} + \frac{\texttt{B sin[t w0]}}{\texttt{w0}} \right\}$

Στην τελευταία εντολή μετατρέψαμε τον ¨κανόνα' σε μια εξίσωση με την αντικατάσταση **Rule → Equal**.

ΕΦΑΡΜΟΓΗ 3 : Βολή σε Σταθερό Βαρυτικό Πεδίο

Εστω ένα σώμα μάζας m σε ένα βαρυτικό πεδίο

1. Να λυθούν οι εξισώσεις κίνησης για τροχιές που έχουν την ίδια αρχική και τελική θέση αλλά διαφορετικούς ολικούς χρόνους
2. Δημιουργήστε γραφήματα των διαφορετικών τροχιών
3. Λύστε τις εξιώσεις κίνησης για τροχιές με συγκεκριμένη αρχική ταχύτητα και γωνία βολής. Να βρεθεί η θέση του μέγιστου ύψους.
4. Δημιούργηστε τα γραφήματα για το προηγούμενο ερώτημα αλλά για διαφορετικές γωνίες.

ΛΥΣΗ
1ο ερώτημα

In[6]:= Clear["Global‘ * "]

Οι εξισώσεις κίνησης για ένα σωματίδιο με μάζα m είναι:

```
In[7]:= eq1 = {m x''[t] == 0, m z''[t] == - m g}
```

όπου z είναι η κάθετη απόσταση (ύψος) και x είναι η οριζόντια απόσταση. Αν η θέση τη χρονική στιγμή $t = 0$ είναι $\{x, y\} = \{0, 0\}$ και η τελική θέση μετά απο χρόνο $t = tf$ είναι $\{xf, zf\}$, οι οριακές συνθήκες είναι:

```
In[8]:= initial1 = {x[0] == 0, x[tf] == xf, z[0] == 0,
          z[tf] == zf}
```

ενώνοντας την **eq1** με την **initial1** και χρησιμοποιώντας την εντολή **DSolve** λαμβάνουμε:

```
In[9]:= dsol1 =
         (DSolve[Join[eq1, initial1], {x[t], z[t]},
              t][[1]]// Simplify)
```

Out[9]= $\left\{x[t] \to \frac{t\ xf}{tf}, z[t] \to \frac{t\,(g\ tf\,(-t + tf) + 2\ zf)}{2\ tf}\right\}$

2ο ερώτημα

Θα εκτυπώσουμε τις τροχιές για τρείς διαφορετικούς τελικούς χρόνους tf=$\{5,10,15\}$ για τις παρακάτω τιμές των παραμέτρων:

```
In[10]:= values1 = {xf → 300, zf → 0, g → 9.8}
```

Οι συντεταγμένες σαν συναρτήσεις του χρόνου είναι

```
In[11]:= coord1[tf_] = {x[t], z[t]}/. dsol1 /. values1
```

Out[11]= $\left\{\frac{300\ t}{tf}, 4.9\ t\,(-t + tf)\right\}$

Η γραφική παράσταση των τροχιών θα δημιουργηθεί με τη χρήση της εντολής **ListPlot**. Θα επιλέξουμε χρονικές υποδιαιρέσεις 0.5 sec.

```
In[12]:= Clear[plot1];
         plot1[tf_] :=
           ListPlot[
             Evaluate[Table[coord1[tf],
                 {t, 0, tf, 0.5}]],
             PlotStyle → PointSize[0.02],
               PlotLabel → (" Same Initial and Final Positions "),
             GridLines → Automatic]
```

```
In[13]:= plotarray2 = Table[plot1[tf], {tf, 5, 15, 5}];
```

```
In[14]:= Show[plotarray2]
```




```
Out[14]= -Graphics-
```

3ο ερώτημα

Θα λύσουμε τις εξισώσεις κίνησης με αρχική ταχύτητα v0 και αρχική γωνία alpha

```
In[15]:= initial2 = {x[0] == 0, x'[0] == v0 Cos[alpha],
           z[0] == 0, z'[0] == v0 Sin[alpha]};
```

Η λύση της **εχ1** με τις παραπάνω αρχικές συνθήκες είναι:

```
In[16]:= dsol2 =
          DSolve[Join[eq1, initial2], {x[t], z[t]},
            t][[1]]
```

Out[16]= $\{x[t] \to t\ v0\ \cos[alpha],\ z[t] \to \frac{1}{2}(-g\,t^2 + 2\,t\ v0\ \sin[alpha])\}$

Ο χρόνος που απαιτείται για να φθάσει στο μέγιστο ύψος υπολογίζεται αν πάρουμε τη χρονική παράγωγο της $z(t)$, μηδενίσουμε το αποτέλεσμα και λύσουμε την εξίσωση ως προς το χρόνο t:

```
In[17]:= eq2 = D[z[t]/. dsol2, t] == 0;
         tsol = Solve[eq2, t][[1]]
```

Out[17]= $\{t \to \frac{v0\ \sin[alpha]}{g}\}$

Η θέση του μέγιστου ύψους λαμβάνεται υπολογίζοντας τα $x[t_{max}]$ $y[t_{max}]$

```
In[18]:= {x[t], z[t]}/. dsol2 /. tsol // Simplify
```

Out[18]= $\{\frac{v0^2\ \cos[alpha]\ \sin[alpha]}{g},\ \frac{v0^2\ \sin[alpha]^2}{2\,g}\}$

4ο ερώτημα

Θα δημιουργήσουμε τις γραφικές παραστάσεις για $alpha = \{\pi/8, 2\pi/8, 3\pi/8,$ χρησιμοποιώντας τις παρακάτω τιμές

```
In[19]:= values2 = {g → 9.8, v0 → 100};
```

για τις παρακάτω συντεταγμένες της τροχιάς:

```
In[20]:= coord2[alpha_] =
           {x[t], z[t]}/. dsol2 /. values2
```

```
Out[20]= {100 t cos[alpha],
           1/2 (−9.8 t^2 + 200 t sin[alpha])}
```

Η γραφική παράσταση θα δημιουργηθεί με τη χρήση της εντολής **ListPlot**. Θα επιλέξουμε χρόνους απο 0 ως 20 seconds με βήματα 1 sec. Θα προσθέσουμε μερικές επι πλεόν επιλογές στη δημιουργία της γραφικής παράστασης που δεν είναι αναγκαίες αλλά βελτιώνουν τη μορφή του γραφήματος.

```
In[21]:= Clear[plot2];
         plot2[alpha_] :=
           ListPlot[
             Evaluate[Table[coord2[alpha],
                 {t, 0, 20, 0.5}]],
             PlotStyle → PointSize[0.02], Frame → True,
             PlotLabel →
               (" Different Initial Angle "),
             GridLines → Automatic,
             FrameLabel → {"x[t]", "z[t]", "Distance",
                 "Height"}]
```

η εντολή **plot2**[**alpha**] θα δημιουργήσει μια τροχιά. Στη συνέχεια μπορούμε να δημιουργησουμε ένα πίνακα με γραφηματα για δοθείσες τιμές του alpha.

```
In[22]:= plotarray2 = Table[plot2[alpha],    {alpha, π/8, 3 π/8, π/8}];
```

```
In[23]:= Show[plotarray2,
           PlotRange → {{0, 1000}, {0, 500}},
           RotateLabel → False];
```



Δεν αποτελεί έκπληξη ότι η τροχία για $alpha = \pi/4$ θα δώσει το μεγαλύτερο βεληνεκές.

0.1.2 Αριθμητική Επίλυση ΔΕ

Αν δεν είναι δυνατή η αναλυτική επίλυση της ΔΕ τότε μπορεί να χρησιμοποιηθεί η εντολή **NDsolve[eqns, y, x, xmin, xmax]** για τον αριθμητικό υπολογισμό των λύσεων. Οπου **εχνς** είναι το σύστημα των ΔΕ για τη συνάρτηση ψ με ανεξάρτητη μεταβλητή **x** στο διάστημα [**xmin, xmax**].

```
In[24]:= eq1 = {x''[t] == -x[t] - x[t]y[t]^2,
             y''[t] == -y[t] - x[t]y[t]^2};
```

με αρχικές συνθήκες

```
In[25]:= initial = {x[0] == 2, x'[0] == 0, y[0] == 1, y'[0] == 0};
```

η αριθμητική λύση του συστήματος των ΔΕ για $t = 1$ ως $t = 20$ δίνεται απο την εντολή:

```
In[26]:= ndsol = NDsolve[Join[eq1, initial], {x, y},
            {t, 0, 20}][[1]]
Out[26]= {x → InterpolatingFunction[{{0., 20.}}, <>],
         y → InterpolatingFunction[{{0., 20.}}, <>]}
```

Τα αριθμητικά δεδομένα δίνονται με τη χρήση της **InterpolatingFunction** της οποίας οι τιμές θα βρεθούν με interpolation (παρεμβολή). Οι γραφικές παραστάσεις είναι:

```
In[27]:= Plot[Evaluate[{x[t], y[t]}/. ndsol],
         {t, 0, 20},
         PlotStyle → {{Dashing[{ }]},
            {Dashing[{0.05}]}}]
```



Out[27]= - Graphics -

Η συνεχής γραμμή είναι η λύση $x[t]$ και η διακεκομένη είναι η $y[t]$. Στο επίπεδο x, y η λύση δίνεται από την εντολή:

In[28]:= **ParametricPlot[**
Evaluate[{x[t], y[t]}/. ndsol], {t, 0, 20}]



Out[28]= - Graphics -

ΠΑΡΑΤΗΡΗΣΕΙΣ

1. Η NDsolve λύνει ένα ευρύ φάσμα ΔΕ αλλά και ορισμένες ΔΕΜΠ
2. η NDsolve δίνει αποτελέσματα μέσω της InterpolatingFunction.
3. Οι παράγωγοι στις ΔΕ πρέπει να ορίζονται ως $y'[x]$

4. Στις κανονικές ΔΕ η συνάρτηση $y[x]$ πρέπει να εξαρτάται μόνο απο μία μεταβλητή. Στις ΔΕΜΠ μπορούν να είναι συναρτήσεις περισσοτέρων μεταβλητών.
5. Οι ΔΕ πρέπει να περιέχουν ικανές αρχικές και οριακές συνθήκες ωστε να μπορεί να επιλυθεί το πρόβλημα.
6. Οι αρχικές και οι οριακές συνθήκες δίνονται συνήθως στη μορφή $y[x_0] == c_0$, $y'[x_0] == dc_0$, κτλπ. Είναι δυνατή και η χρήση πιο σύνθετων οριακών συνθηκών.
7. Περιοδικές ορικές συνθήκες μπορούν να ορισθούν μέσω της εξίσωσης $y[x_0] == y[x_1]$.
8. Οι ΔΕ στην NDsolve μπορούν να έχουν και μιγαδικές μεταβλητές.
9. Οι παρακάτω επιλογές μπορουν να δοθούν:
   - AccuracyGoal
   - InterpolatingPrecision
   - MaxSteps (χρησιμοποιεί αυτόματα 1000)
   - MaxStepSize
   - Method (για συνήθη ΔΕ χρησιμοποιεί τη μέθοδο Adams. Υπάρχουν και οι επιλογές Gear, Runge-Kutta (4-5ης τάξης Runge-Kutta-Fehlberg))
   - PrecisionGoal
   - StartingStepSize
   - WorkingPrecision

ΕΦΑΡΜΟΓΗ 4: Συζευγμένοι Αρμονικοί Ταλαντωτές

Οι εξισώσεις κίνησιες δύο συζευγμένων αρμονικών ταλαντωτών (δύο εκκρεμή που συνδέονται με ένα ελατήριο) δίνονται απο τις σχέσεις

$$\begin{aligned} M\frac{d^2x}{dt^2} &= -c_1x + c(y-x) \\ M\frac{d^2y}{dt^2} &= -c_2y - c(y-x) \end{aligned}$$

οι σταθερές c_1, c_2 υπολογίζονται απο τα μήκη των εκκρεμών και τη βαρυτική έλξη Mg ενώ c είναι η σταθερά του ελατηρίου.

θα υποθέσουμε $c_1 = c_2$ και θα εξετάσουμε δύο περιπτώσεις μία για ασθενή σύζευξη ($c = c_1/10$) και μία για ισχυρή σύζευξη ($c = 10c_1$).

ΛΥΣΗ

```
In[29]:= eq = {M x''[t] == - C1 x[t] + C(y[t] - x[t]),
           M y''[t] == - C2 y[t] - C(y[t] - x[t])};
```

δίνουμε τις παρακάτω αρχικές τιμές (σε m)

```
In[30]:= initial = {x[0] == 5, x'[0] == 0, y[0] == 8,
           y'[0] == 0};
```

και δύο ζεύγη παραμέτρων c_1, c_2, c για ασθενή και ισχυρή σύζευξη (σε μονάδες Nm^{-1})

```
In[31]:= values1 = {C1 → 39.5, C2 → 39.5, C → 3.95,
           M → 1};
         values2 = {C1 → 39.5, C2 → 39.5, C → 395,
           M → 1};
```

```
In[32]:= eq1 = eq /. values1; eq2 = eq /. values2;
```

```
In[33]:= ndsol1 =
          NDsolve[Join[eq1, initial], {x, y},
             {t, 0, 5}, MaxSteps → 1000][[1]];
         ndsol2 =
          NDsolve[Join[eq2, initial], {x, y},
             {t, 0, 5}, MaxSteps → 4000][[1]];
```

```
In[34]:= Plot[Evaluate[{x[t], y[t]}/. ndsol1],
           {t, 0, 5},
           PlotStyle → {{Dashing[{ }]},
              {Dashing[{0.01}]}},
           PlotLabel − > "Weak Coupling"]
          Plot[Evaluate[{x[t], y[t]}/. ndsol2],
           {t, 0, 5},
           PlotStyle → {{Dashing[{ }]},
              {Dashing[{0.01}]}},
           PlotLabel − > "Strong Coupling"]
```





Out[34]= (- Graphics -) (- Graphics -)

```
In[35]:= ParametricPlot[
       Evaluate[{x[t], y[t]}/. ndsol1],
       {t, 0, 5}, PlotLabel -> "Weak Coupling"]
     ParametricPlot[
       Evaluate[{x[t], y[t]}/. ndsol2],
       {t, 0, 5}, PlotLabel -> "Strong Coupling"]
```





Out[35]= (-Graphics-) (-Graphics-)

Αξιζει να δοκιμάσουμε διάφορες ομάδες αρχικών τιμών και παραμέτρων.

ΕΦΑΡΜΟΓΗ 5: Αρμονικος Ταλαντωτής με Απόσβεση

Ας θεωρήσουμε ένα αρμονικό ταλαντωτή με απόσβεση. Η εξίσωση κίνησης θα είναι:

$$x''(t) + \gamma x'(t) + \omega_0^2 x(t) = 0$$

Η απόσβεση είναι ανάλογη της ταχύτητας και χαρακτηρίζεται απο την παράμετρο γ.

1. Να λυθεί η εξίσωση με αρχικές συνθήκες $x(0) = x_0$ και $x'(0) = u_0$. Στη συνέχεια να γίνει το γράφημα της λύσης σαν συνάρτηση του t και του $2\omega = \sqrt{\gamma^2 - 4\omega_0^2}$.
2. Να γίνει το γράφημα για την φθίνουσα περίπτωση, (πραγματικό ω ή $\gamma^2 > 4\omega_0^2$).
3. Να γίνει το γράφημα για την οριακά φθίνουσα περίπτωση, ($\omega = 0$ ή $\gamma^2 = 4\omega_0^2$).
4. Να γίνει το γράφημα για την περίπτωση της φθίνουσας ταλάντωσης, (φανταστικό ω ή $\gamma^2 < 4\omega_0^2$). Δημιούργησε διαγράμματα φάσης αλλά και χρόνου-φάσης.
5. Σχεδίασε όλες τις παραπάνω λύσεις σε ένα γράφημα.

ΛΥΣΗ

1ο ερώτημα

In[36]:= Clear["Global`*"];

Θα θέσουμε $\omega = w$ και $\gamma = gam$ οπότε η ΔΕ ορίζεται ως:

```
In[37]:= eq1 = x''[t] + gam x'[t] + w0^2 x[t] == 0;
```

θα τη λύσουμε με τη βοήθεια της NDsolve:

```
In[38]:= dsol =
           DSolve[{eq1, x[0] == x0, x'[0] == v0},
              x[t], t][[1]];
```

Πριν παρουσιάσουμε τη λύση είναι χρήσιμο να κάνουμε την αντικατάσταση $2\omega = \sqrt{\gamma^2/4 - \omega_0^2}$

```
In[39]:= wsub = {(gam^2 - 4 * w0^2)^(1/2) → (2 w),
            1/(gam^2 - 4 * w0^2)^(1/2) → 1/(2 w)};
```

οπότε οι λύση είναι:

```
In[40]:= dsol = (dsol // Simplify) //. wsub;
         x[t_] = x[t] /. dsol
```

Out[40]= $\frac{1}{4\,w}\left(e^{-\frac{1}{2}\,t\,(gam+2\,w)}\,(2\,(-1+e^{2\,t\,w})\,v0+((-1+e^{2\,t\,w})\,gam+2\,(1+e^{2\,t\,w})\,w)\,x0)\right)$

είναι προφανές απο την παραπάνω λύση ότι έχουμε τρείς διαφορετικές λύσεις ανάλογα με τις τιμές του $2\omega = \sqrt{\gamma^2 - 4\omega_0^2}$, δηλαδή αν είναι πραγματική, μηδέν ή φανταστική. Και οι τρείς περιπτώσεις θα μελετηθούν στη συνέχεια.

```
In[41]:= values = {x0 → 1, v0 → 0, gam → 1};
```

Παρατήρηση: Η αντικατάσταση $\omega \to \sqrt{\omega^2}$ γίνεται ωστε το ω να γίνεται μιγαδικό οταν θέτουμε αρνητικές τιμές στο $w2$. Μεταβάλλοντας το $w2$ απο αρνητικές τιμες στο μηδέν και στη συνέχεια σε θετικές τιμές λαμβάνουμε τα τρία είδη λύσεων που προαναφέραμε.

```
In[42]:= p0 =
          Plot3D[
            Evaluate[ x[t]//. values /.
                {w → Sqrt[w2]}], {w2, 25, −150},
            {t, 0, 1}, PlotPoints → 25,
            PlotRange → {−1, 2}, BoxRatios → {1, 1, 1},
            ViewPoint → {−4, 3, 2},
            AxesLabel → {"w2", "t", "x[t]"}]
```




```
Out[42]= - SurfaceGraphics -
```

Είναι προφανές πως όταν το ω2 λαμβάνει μεγάλες αρνητικές τιμές η λύση ξ[τ] παρουσιάζει απόσβεση. ω2=0 αντιστοιχεί στην οριακή περίπτωση και για 0˘2ω˘γαμ η λύση αποσβένυται εκθετικά. Για γαμ˘2ω η λύση αυξάνεται εκθετικά.

2ο ερώτημα

Για την περίπτωση που το ω είναι πραγματικό $\gamma^2 > 4\omega_0^2$ η λύση είναι εκθετικά αύξουσα $(2\omega > 0)$ ή φθίνουσα $(0 < 2\omega < \gamma)$, όπως φαίνεται στα παρακάτω γραφήματα.

```
In[43]:= p1 =
           Plot[Evaluate[x[t]//. values /.{w → 0.6}],
             {t, 0, 10}]
```




```
Out[43]= -Graphics-
```

```
In[44]:= p2 =
           Plot[Evaluate[x[t]//. values /.{w → 0.4}],
             {t, 0, 10}]
```




```
Out[44]= -Graphics-
```

3ο ερώτημα

Η οριακή περίπτωση ω=0 ($\gamma^2 = 4\omega_0^2$) είναι εξαιρετικά απλή:

```
In[45]:= Limit[x[t], w → 0]
```

Out[45]= $\frac{1}{2} e^{-\frac{\text{gam t}}{2}} (2 \text{ t v0} + 2 \text{ x0} + \text{gam t x0})$

```
In[46]:= p3 =
    Plot[
      Evaluate[Limit[x[t]//. values, w → 0]],
      {t, 0, 10}]
```




```
Out[46]= - Graphics -
```

4ο ερώτημα

Σε αυτή την περίπτωση το ω είναι φανταστικό ($\gamma^2 < 4\omega_0^2$).

```
In[47]:= (x[t]// Simplify)
```

Out[47]= $\frac{1}{4\,w}\left(e^{-\frac{1}{2}\,t\,(gam+2\,w)}\,(2\,(-1+e^{2\,t\,w})\,v0+((-1+e^{2\,t\,w})\,gam+2\,(1+e^{2\,t\,w})\,w)\,x0)\right)$

Η λύση προφανώς είναι μια ταλάντωση με συχνότητα ω και εκθετική απόσβεση $e^{-\gamma t/2}$, όπως φαίνεται στο παρακάτω γράφημα:

```
In[48]:= p4 =
    Plot[Evaluate[x[t]//. values /.{w → i 10}],
      {t, 0, 10}]
```



Out[48]= - Graphics -

Χρήσιμα για τη μελέτη μας είναι το διαγράμμα φάσης, αλλά και η εξέλιξη της φάσης ως συνάρτηση του χρόνου. Τα παρακάτω γραφήματα εξυπηρετούν αυτό το σκοπό:

In[49]:= **q = x[t]//. values /.{w → i 10} // Simplify**

Out[49]= $\frac{1}{40}\, e^{\left(-\frac{1}{2}-10\, I\right) t} \left((20+I)+(20-I)\, e^{20\, I\, t}\right)$

In[50]:= **p = x′[t]//. values /.{w → i 10}// Simplify**

Out[50]= $\frac{401}{80}\, I\, e^{\left(-\frac{1}{2}-10\, I\right) t} \left(-1+e^{20\, I\, t}\right)$

In[51]:=
```
ParametricPlot[{p, q}, {t, 0, 10},
  PlotRange → All, Axes → False,
  PlotLabel − > "Phase Trajectory",
  Frame → True,
  FrameLabel → {"p − axis", "q − axis"}]
```



Out[51]= - Graphics -

```
In[52]:= ParametricPlot3D[{p, q, t}, {t, 0, 10},
           PlotRange → All, Axes → False,
           PlotLabel − > "Evolution of Phase",
           PlotPoints → 300, BoxRatios → {1, 1, 1},
           AxesLabel → {"p", "q", "t"}]
```




```
Out[52]= - Graphics3D -
```

5ο ερώτημα
Συνολικά όλες οι λύσεις συγχρονως είναι:

```
In[53]:= Show[p1, p2, p3, p4]
```



Out[53]= -Graphics-

Πολλές φορές είναι χρήσιμο να υπολογίσουμε τη συχνότητα μια ταλάντωσης και προς τούτο χρησιμοποιούμε το μετασχηματισμό Fourier. Παρακάτω φαίνοντα τα βήματα που πρέπει να ακολουθήσουμε.

```
In[54]:= fdata = Table[p, {t, 0, 40, 0.001}]// N;

In[55]:= ffData = Abs[Fourier[fdata]];

In[56]:= ListPlot[ffData, PlotJoined → True,
           PlotRange -> {{0, 120}, All}]
```



Out[56]= -Graphics-

Στο παραπάνω διάγραμμα παρατηρούμε ότι το μέγιστο είναι για $2\pi\omega \approx 62.8$.

ΝΑ ΛΥΣΕΤΕ ΤΟ ΠΑΡΑΠΑΝΩ ΠΡΟΒΛΗΜΑ ΜΕ ΤΗ ΧΡΗΣΗ ΤΗΣ NDsolve, ΑΥΤΟ ΘΑ ΣΑΣ ΒΟΗΘΗΣΕΙ ΝΑ ΛΥΣΕΤΕ ΤΑ ΠΡΟΒΛΗΜΑΤΑ ΣΤΟ ΤΕΛΟΣ

ΕΦΑΡΜΟΓΗ 6: Το πρόβλημα Kepler

Η εξίσωση κίνησης για ένα σωματίδιο που κινείται σε ένα πεδίο κεντρικών δυνάμεων μπορεί να γραφεί ώς:

$$u''(\phi) = -u(\phi) - \frac{m}{\ell^2}\frac{f(1/u(\phi))}{u(\phi)^2}$$

όπου $u = 1/r$. Υποθέτουμε πως η τροχιά είναι στο ισημερινό επίπεδο ($\theta = \pi/2$) και ότι $u = u(\phi)$.

1. Να λυθεί η παραπάνω εξίσωση για το βαρυτικό δυναμικό $V = -k/r$ με αρχικές συνθήκες $u'(0) = 0$.
2. Να γίνουν γραφήματα των ελλειπτικών και υπερβολικών τροχιών.

ΛΥΣΗ

1ο ερώτημα

```
In[57]:= Clear["Global`*"];

In[58]:= Needs["Graphics`Graphics`"]

In[59]:= eq1 = u''[φ] ==
             -u[φ] - m f[1/u[φ]]/(L^2 u[φ]^2);
```

Η δύναμη δίνεται ώς η παράγωγος του δυναμικού $f(r) = -V'(r)$. Στη συνέχεια δημιουργούμε μια εντολή που υπολογίζει τη δύναμη ως συνάρτηση του δυναμικού (μπορεί να χρησιμοποιειθεί και για πλεόν σύνθετα δυναμικά).

```
In[60]:= forceRule =
           {V → (-k/#&), f[1/u[φ]] → -V'[r],
            r → 1/u[φ]};
```

αντικαθιστώ το βαρυτικό δυναμικό στην εχ1:

```
In[61]:= eq2 = eq1 //. forceRule
```

Out[61]= $\left(u''[\phi] == \frac{k\,m}{L^2} - u[\phi]\right)$

Για λόγους απλότητας θα χρησιμοποιήσουμε την αρχική συνθήκη $u'(0) = 0$. Εφαρμόζοντας την ΔΣολε στην εχ2 λαμβάνουμε

```
In[62]:= eq3 =
           DSolve[{eq2, u'[0] == 0, u[0] == u0(1 + e)},
              {u[φ]}, φ][[1]]// Simplify
```

Out[62]= $\left\{u[\phi] \to \frac{k\,m + (-k\,m + (1+e)\,L^2\,u0)\cos[\phi]}{L^2}\right\}$

Η γενική μορφή της εξίσωσης της τροχιάς μπορεί να γραφεί ως $u(\phi) = u_0[1 + e\cos(\phi)]$. Οταν $\phi = 0$ μπορούμε να λύσουμε για $u0$:

```
In[63]:= u0Rule =
           Solve[(u[φ]/. eq3 /.{cos[φ] → 0}) ==
              u0, u0][[1]]
```

Out[63]= $\left\{\text{u0} \to \frac{\text{k m}}{\text{L}^2}\right\}$

Αντικαθιστώντας την παραπάνω σχέση στην eq3, καταλήγουμε στη λύση:

In[64]:= **eq4 = eq3 //. u0Rule // Simplify**

Out[64]= $\left\{\text{u}[\phi] \to \frac{\text{k m }(1+\text{e cos}[\phi])}{\text{L}^2}\right\}$

Η παράμετρος e (εκκεντρότητα) καθορίζει τα τρία είδη των κωνικών τομών: (1) $e > 1$ υπερβολικήες, (2) $e < 1$ ελλειπτικές ($e = 0$ κυκλική) και (3) $e = 1$, παραβολικές.

Συνηθίζεται να ορίζουμε την ποσότητα a που ορίζεται ως $\ell^2/(km) = a(1 - e^2)$, οπότε η u[phi] γράφεται:

In[65]:= **eq5 = eq4 //. {L → Sqrt[a k m (1 − e^2)]}**

Out[65]= $\left\{\text{u}[\phi] \to \frac{1+\text{e cos}[\phi]}{\text{a }(1-\text{e}^2)}\right\}$

2ο ερώτημα

θα χρησιμοποιήσουμε την εντολή **PolarPlot** για να δημιουργήσουμε τις γραφικές παραστάσεις των ελλεπτικών και υπερβολικών τροχιών

```
In[66]:= Clear[plotOrbit];
         plotOrbit[aa_, ee_] :=
           PolarPlot[
             (1/u[φ] //. eq5  //.{a → aa, e → ee} //
                Evaluate ), {φ, 0, 4 π}];
```

Ελλεπτικές τροχιές

```
In[67]:= Show[
           Table[plotOrbit[1, eIn],
              {eIn, 0.1, 0.9, 0.2}]// Evaluate]
```



Out[67]= - Graphics -

Υπερβολικές τροχιές

```
In[68]:= plotOrbit[1, 1.1]
```




```
Out[68]= -Graphics-
```

```
In[69]:= plotOrbit[1, 1.7]
```




```
Out[69]= -Graphics-
```

ΝΑ ΛΥΣΕΤΕ ΤΟ ΠΑΡΑΠΑΝΩ ΠΡΟΒΛΗΜΑ ΜΕ ΤΗ ΧΡΗΣΗ ΤΗΣ NDsolve, ΑΥΤΟ ΘΑ ΣΑΣ ΒΟΗΘΗΣΕΙ ΝΑ ΛΥΣΕΤΕ ΤΑ ΠΡΟΒΛΗΜΑΤΑ ΣΤΟ ΤΕΛΟΣ

0.2 ΑΛΥΤΑ ΠΡΟΒΛΗΜΑΤΑ

1. Επαναλάβετε το λυμένο παράδειγμα με τις βολές προσθέτωντας την αντίσταση του αέρα που θα είναι $-bu$ για $0 \leq b \leq 1$.

2. Ενα βλήμα εκτοξεύεται με αρχική ταχύτητα $20m/sec$ με γωνία 30^0. Λίγο αργότερα διασπάται σε δύο τμήματα ένα εκ των οποίων έχει διπλάσια μάζα απο το δεύτερο. Τα δύο βλήματα προσγείωνονται συγχρόνως. Το ελαφρύτερο σε απόσταση $20m$ απο το σημείο εκτόξευσης προς τη διεύθυνση που εκτοξεύτηκε αρχικά το βλήμα.

   - Που προσγειώνεται το έτερο βλήμα·
   - Δημιούργησε τις γραφικές παραστάσεις των δύο τροχιών, αλλά και αυτή του κέντρου μάζης.
   - Δημιούργησε μια συνάρτηση που θα δημιουργεί το γράφημα της κίνησης των δύο τμημάτων και του κέντρου μάζης όταν δίνονται: η αρχική ταχύτητα, η μάζα των τμημάτων και το σημείο πρόσπτωσης του ενός από τα τμήματα.

3. Στο πρόβλημα των συζευγμένων αρμονικών ταλαντωτών δοκιμάστε για τις ίδιες αρχικές τιμές τις τιμές $c_1 = 40\,Nm^{-1}$, $c_2 = 30\,Nm^{-1}$ και $c = 10\,Nm^{-1}$

4. Θεωρήστε το πρόβλημα του αρμονικού ταλαντωτή με μια εξωτερική δύναμη διέγερσης για τον οποίον η εξίσωση κίνησης γράφεται ως

$$x''[t] + \gamma x'[t] + \omega_0^2 x[t] = Q_0 \cos(w_d t)$$

   όπου $4\omega_0^2 > \gamma$. Η λύση θα είναι η υπέρθεση της λύσης της ομογενούς ΔΕ και μιας μερικής λύσης.

   (α′) Βρείτε τη λύση της ομογενούς ΔΕ.

   (β′) Υποθέστε ότι η μερική λύση είναι της μορφής $A\cos(\omega_d t + \delta_f)$ και υπολογίστε τα A και δ_f. Γράψτε τη γενική μορφή της λύσης.

   (γ′) Δημιουργήστε γραφικές παραστάσεις της λύσης ως συνάρτηση του ω_0 και του t. Δημιουργήστε γραφικές παραστάσεις του A ως συνάρτηση του ω_0 και του γ. Παρατηρήστε τους συντονισμούς για $\omega_0 = \omega_d$.

   (δ′) Δημιουργήστε διαγράμματα φάσης.

5. Θεωρήστε τη μη-γραμμική ΔΕ

$$q''(t) - aq'(t) + bq^3(t) = 0$$

   υποθέστε ότι $b > 0$ και a θετικό ή αρνητικό. Η εξίσωση αυτή είναι γνωστή ώς *εξίσωση Duffing*.

   (α′) Λύστε αριθμητικά την ΔΕ και δημιουργήστε τις γραφικές παραστάσεις των $q(t)$ και $q'(t)$. Δώστε αρχικές τιμές π.χ. $q(0) = 0$ και $q'(0) = 0.001$. Υποθέστε για τις παραμέτρους τα παρακάτω ζεύγη τιμών $(b = 0.05, a = -1)$ $(b = 0.05, a = 4)$. Τι παρατηρείτε για κάθε ένα απο τα ζεύγη των παραμέτρων.

   (β′) Δημιουργήστε διαγράμματα φάσης για κάθε μια απο τις παραπάνω περιπτώσεις.

   (γ′) Δημιουργήστε προσομοιώσεις των παραπάνω περιπτώσεων.

   (δ′) (Προαιρετικό πρόβλημα) Θα μπορροούσατε να δημιουργήσετε γραφηματα του δυναμικού για την παραπάνω εξίσωση σαν συνάρτηση του q και να εξετάσετε την εξάρτηση απο το a·

6. Θεωρήστε τη μη-γραμμική ΔΕ

$$q''(t) = aq(t) - bq^3(t) - \gamma q'(t) + Q_0 \cos(\omega t)$$

με αρχικές συνθήκες $q(0) = 0$ $q'(0) = 0.001$. Υποθέστε $a = 0.4$, $b = 0.5$, $\gamma = 0.2$ και $\omega = 1/8$ και θεωρήστε δύο τροχίες μία για $Q_0 = 0$ και μια για $Q_0 = 0.1$.

   (α΄) Να λύσετε αριθμητικά τις εξισώσεις κίνησης και για τις δύο παραπάνω περιπτώσεις και να δημιουργήσετε γραφικές παραστάσεις για τα $q(t)$ και $q'(t)$.

   (β΄) Δημιουργήστε τα διαγράματα φάσης, και χρησιμοποιήστε το μετασχηματισμό Fourier για τον υπολογισμό της συχνότητας.

7. **Γενικευμένο πρόβλημα Kepler**. Το δυναμικό δίνεται απο τη σχέση $V = -k/r - b/r^2$

   (α΄) Λύστε την εξίσωση Kepler για $u[\phi]$ και εκφράστε τη λύση στη μορφή $u(\phi) = u_0(1 + e\cos[\phi(1-\delta)])$ όπου $u_0 = k\ m/(\ell^2 - 2\ b\ m)$.

   (β΄) Δημιουργήστε γραφικές παραστάσεις και εξετάστε τις ελλειπτικές τροχιές, τι παρατηρήτε·

   (γ΄) Συγκρίνετε τις τροχιές των δυναμικών $V = -k/r - b/r^3$ και $V = -k/r + b/r^3$.

   (δ΄) Συγκρίνετε τις τροχιές των δυναμικών $V = -k/r^{1.1}$ και $V = -k/r^{0.9}$.

8. Ξρησιμοποιείστε τη μέθοδο Runge -Kutta-Felberg για την αριθμητική επίλυση των παρακάτω διαφορικών εξισώσεων και συγκρίνετε με τις ακριβείς τιμές:

   (α΄) $y' = xy^{1/3}$, με $y(1) = 1$ (ακριβής $y = \left(\frac{x^2+2}{3}\right)^{3/2}$).

   Υπόδειξη: ολοκληρώστε από $x = 1$ ως $x = 2$

   (β΄) $y' = -xy^2$ με $y\,(1) = 2$ (ακριβής $y = \frac{2}{x^2}$)

   Υπόδειξη: ολοκληρώστε από $x = 1$ ως $x = 2$

   (γ΄) $y' = -2xy$ με $y\,(0) = 1$ (ακριβής $y = e^{-x^2}$)

   Υπόδειξη: ολοκληρώστε από $x = 0$ ως $x = 1$

   (δ΄)

$$y' = \frac{y\left(1 - x^2y^4\right)}{x\left(1 + x^2y^4\right)}$$

   με $y\,(1) = 1$.

   Υπόδειξη: ολοκληρώστε από $x = 1$ ως $x = 2$

   (ε΄)

$$y' = \frac{x - e^x}{y + e^y}$$

   με $y\,(0) = 0$. Βρείτε το $y\,(1)$.

9. Να λυθεί αριθμητικά το σύστημα διαφορικών εξισώσεων

$$\begin{aligned} x' &= 1195x - 1995y \qquad x(0) = 2 \\ y' &= 1197x - 1997y \qquad y(0) = -2 \end{aligned}$$

Να βρεθούν: οι τιμές $y(1)$, $x(1)$ και οι τιμές $y(-1)$, $x(-1)$ Τι παρατηρείτε· αν η ακριβής λύση είναι:

$$x(t) = 10e^{-2t} - 8e^{-800t}, y(t) = 6e^{-2t} - 8e^{-800t}$$

Δοκιμάστε να χρησιμοποιήσετε τη μέθοδο Gear στην NDsolve, και συγκρίνετε με την μέθοδο Runge-Kutta και Adams τι παρατηρείτε·.

10. Να λυθούν τα συστήματα:

$$\begin{aligned} y' &= yz + x \qquad y(0) = 0 \\ z' &= y - x \qquad z(0) = 0 \end{aligned}$$

και

$$\begin{aligned} y' &= z \qquad y(0) = 0 \\ z' &= -y \qquad z(0) = 1 \end{aligned}$$